

```

// Hauptauftrag: // geh
lerne neu
// initialisiert die Parameter mangels des Bildschirms, vom Stift und vom Igel. //
bild freimacht beschleunige fi grün stiftdicke 1 verstecktigel
ende

```

```

lerne Spire
setze "l xyKurve 0 1
setze "x1 erstes :l setze "y1 letztes :l
wiederhole 360 [
    fi FarbWinkel whzahl
    setze "l xyKurve whzahl :Fall //:Fall = 1, 2 ou 3 //
    setze "x erstes :l setze "y letztes :l
    segment satz :x1 :y1 satz :x :y
    setze "x1 :x
    setze "y1 :y
]
ende

```

```

lerne zufall
startezufall 1 5
setze "r element [80 90 100 110 120] zufallzahl
setze "s 160 -:r startezufall 0 9
setze "a (1 + zufallzahl )
setze "b (1 + zufallzahl )
setze "c (1 + zufallzahl )
setze "d (1 + zufallzahl )
ende

```

```

lerne FarbWinkel :theta
startezufall 1000 32500
rückgabe zufallzahl
ende

```

```

lerne xyKurve :t :Fall //:Fall = 1, 2 ou 3 //
prüfe :Fall =1
wennwahr [
    setze "x :r *(sin :a *:t) + :s *(sin :b *:t)
    setze "y :r *(cos :c *:t) + :s *(cos :d *:t)
    rückgabe satz :x :y
]
wennfalsch [
    prüfe :Fall = 2
    wennwahr [ // Spiren //
        setze "x :r *(cos :a *:t) + :s *(sin :b *:t)
        setze "y :r *(sin :c *:t) + :s *(cos :d *:t)
        rückgabe satz :x :y
    ]
    wennfalsch [ // Blumen ( Die Parameter: c und: d werden nicht benutzt. //
        setze "x :r *(cos :a *:t) + :s *(sin :b *:t)
        setze "y :r *(sin :a *:t) + :s *(cos :b *:t)
        rückgabe satz :x :y
    ]
]
ende

```

```

lerne geh
teil [ "a "b "c "d "r "s "t "Fall]
löschttext setze "Fall 1
wiederhole 3 [
    schreiblinie :Fall
    wiederhole 12 [neu zufall Spire wart 0,25]
    setze "Fall :Fall +1
]
ende

```